

Unidad 1: Conceptos fundamentales de Programación

Helder Octavio Fernández Guzmán

Objetivos

- Comprender qué es un algoritmo y cómo se representa.
- Conocer el pseudocódigo como herramienta de diseño.
- Identificar variables y tipos de datos.
- Usar operadores básicos.
- Aplicar estructuras de control: condicionales y ciclos.

¿Qué es un algoritmo?

- Conjunto ordenado y finito de pasos.
- Permite resolver un problema o realizar una tarea.
- Debe ser preciso, definido y tener un resultado.

Pseudocódigo y formalidad

- Puente entre lenguaje natural y lenguaje de programación.
- Permite expresar soluciones sin ambigüedad.
- Ejemplo breve:

```
Algoritmo SumaSimple
  Definir a, b, suma Como Entero
  Leer a
  Leer b
  suma <- a + b
  Escribir "La suma es: ", suma
FinAlgoritmo
```

Tipos de instrucciones

- **Entrada:** Leer
- **Salida:** Escribir
- **Asignación:** <-

Variables y tipos de datos

- **Variable:** espacio de memoria con nombre.
- **Tipos comunes:**
 - **Entero:** números sin decimales.
 - **Real:** números con decimales.
 - **Cadena:** texto.
 - **Lógico:** *Verdadero / Falso*.
 - **Caracter:** un solo símbolo.

Operadores (resumen)

Asignación	<- (guarda el valor de la derecha en la variable de la izquierda)
Aritméticos	+ suma, - resta, * multiplicación, / división, % módulo
Relacionales	= igual, <> distinto, > mayor, < menor, >= mayor o igual, <= menor o igual
Lógicos	Y (AND), O (OR), NO (NOT)

Precedencia (de mayor a menor): paréntesis → aritméticos → relacionales → lógicos.

Operadores (ejemplo breve)

```
Algoritmo DemoOperadores
  Definir a, b Como Entero
  Definir r Como Real
  Leer a
  Leer b
  r <- (a + b) / 2 // aritmticos +, /
  Si (a > b) Y (b <> 0) Entonces // relacionales y lgicos
    Escribir "a/b = ", a / b
  SiNo
    Escribir "Condicin no vlida"
  FinSi
  Escribir "Promedio = ", r
FinAlgoritmo
```


Condicional simple

```
Algoritmo MayorEdad
  Definir edad Como Entero
  Leer edad
  Si edad >= 18 Entonces
    Escribir "Es mayor de edad"
  FinSi
FinAlgoritmo
```

Condicional doble

```
Algoritmo ParImpar
  Definir n Como Entero
  Leer n
  Si  $n \% 2 = 0$  Entonces
    Escribir "Es par"
  SiNo
    Escribir "Es impar"
  FinSi
FinAlgoritmo
```

Condicional múltiple

```
Algoritmo MenuOpciones
  Definir opcion Como Entero
  Escribir "1. Sumar"
  Escribir "2. Restar"
  Leer opcion
  Segun opcion Hacer
    1: Escribir "Elegiste Sumar"
    2: Escribir "Elegiste Restar"
  De Otro Modo:
    Escribir "Opcin invlida"
  FinSegun
FinAlgoritmo
```

Bucle Mientras

```
Algoritmo CuentaAscendente
  Definir i Como Entero
  i <- 1
  Mientras i <= 5 Hacer
    Escribir i
    i <- i + 1
  FinMientras
FinAlgoritmo
```

Bucle Repetir

```
Algoritmo AdivinaNumero
  Definir n Como Entero
  Repetir
    Leer n
  Hasta Que n = 7
  Escribir "Adivinaste!"
FinAlgoritmo
```

Bucle Para

```
Algoritmo TablaMultiplicar
  Definir i Como Entero
  Para i <- 1 Hasta 10 Con Paso 1 Hacer
    Escribir "5 x ", i, " = ", 5 * i
  FinPara
FinAlgoritmo
```

Ejercicio integrador

```
Algoritmo PromedioNotas
  Definir i, n Como Entero
  Definir nota, suma Como Real
  suma <- 0
  Escribir "Cantidad de notas: "
  Leer n
  Para i <- 1 Hasta n Hacer
    Leer nota
    suma <- suma + nota
  FinPara
  Si (suma/n) >= 51 Entonces
    Escribir "Aprobado"
  SiNo
    Escribir "Reprobado"
  FinSi
FinAlgoritmo
```

- Algoritmos y formalidad del pseudocódigo.
- Variables, tipos de datos y **operadores**.
- Estructuras de control: condicionales y ciclos.
- Ejercicio integrador (promedio de notas).